RECHNEN MIT MATRIZEN

```
> restart:
> with(LinearAlgebra):
```

Eingabe von Matrizen

```
> Matrix(3,2,[1,2,3,4,5,6]);
```

$$\begin{bmatrix} 1 & 2 \\ 3 & 4 \\ 5 & 6 \end{bmatrix} \quad (1)$$

`> M:=Matrix([[1,2,3,4],[4,5,6,7]]);` (diese Matrix kann im weiteren mit dem Namen M aufgerufen werden)

$$M := \begin{bmatrix} 1 & 2 & 3 & 4 \\ 4 & 5 & 6 & 7 \end{bmatrix} \quad (2)$$

`> M[1,2];` (der Koeffizient der Matrix M in der ersten Zeile und zweiten Spalte)

$$2 \quad (3)$$

`> M[1,2]:=3;` (der Eintrag in der 1. Zeile und 2. Spalte von M wird geändert)

$$M_{1,\,2} := 3 \quad (4)$$

```
> M;
```

$$\begin{bmatrix} 1 & 3 & 3 & 4 \\ 4 & 5 & 6 & 7 \end{bmatrix} \quad (5)$$

```
> Q:=Matrix(2,4,(i,j)->i-j+7);
```

$$Q := \begin{bmatrix} 7 & 6 & 5 & 4 \\ 8 & 7 & 6 & 5 \end{bmatrix} \quad (6)$$

```
> N:=<<1|2|3>,<3|4|5>>;
```

$$N := \begin{bmatrix} 1 & 2 & 3 \\ 3 & 4 & 5 \end{bmatrix} \quad (7)$$

```
> N:=<<1,3>|<2,4>|<3,5>>;
```

$$N := \begin{bmatrix} 1 & 2 & 3 \\ 3 & 4 & 5 \end{bmatrix} \quad (8)$$

```
> v:=Vector([1,2,3,4]);
```

$$v := \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \end{bmatrix} \quad (9)$$

```
> w:=Vector[row]([1,2,3,4]);
```

$$w := \begin{bmatrix} 1 & 2 & 3 & 4 \end{bmatrix} \quad (10)$$

```
> v:=<1,2,3,4>;
```

$$v := \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \end{bmatrix} \quad (11)$$

```
> w:=<1|2|3|4>;
```

$$w := \begin{bmatrix} 1 & 2 & 3 & 4 \end{bmatrix} \quad (12)$$

```
> I24:=IdentityMatrix(2,4);
```

$$I24 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix} \quad (13)$$

Addition von Matrizen, Multiplikation von Matrizen mit Zahlen

```
> M; 2*I24+3*M;
```

$$\begin{bmatrix} 1 & 3 & 3 & 4 \\ 4 & 5 & 6 & 7 \end{bmatrix}$$

$$\begin{bmatrix} 5 & 9 & 9 & 12 \\ 12 & 17 & 18 & 21 \end{bmatrix} \quad (14)$$

```
> 2+3*M;
```

(Bedeutet: Addiere 2*I24 und M)

$$\begin{bmatrix} 5 & 9 & 9 & 12 \\ 12 & 17 & 18 & 21 \end{bmatrix} \quad (15)$$

Multiplikation von Matrizen

```
> M; v; M.v;
```

$$\begin{bmatrix} 1 & 3 & 3 & 4 \\ 4 & 5 & 6 & 7 \end{bmatrix}$$

$$\begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \end{bmatrix}$$

$$\begin{bmatrix} 32 \\ 60 \end{bmatrix} \tag{16}$$

```
> w; v; w.v;
```

$$\begin{bmatrix} 1 & 2 & 3 & 4 \end{bmatrix}$$

$$\begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \end{bmatrix}$$

$$30 \tag{17}$$

```
> v; w; v.w;
```

$$\begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 3 & 4 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 2 & 3 & 4 \\ 2 & 4 & 6 & 8 \\ 3 & 6 & 9 & 12 \\ 4 & 8 & 12 & 16 \end{bmatrix} \tag{18}$$

```
> P:=<<1,2>|<3,4>>;
```

$$P := \begin{bmatrix} 1 & 3 \\ 2 & 4 \end{bmatrix} \tag{19}$$

```
> P^2;
```

$$\begin{bmatrix} 7 & 15 \\ 10 & 22 \end{bmatrix} \tag{20}$$

```
> P^20;
```

$$\begin{bmatrix} 95799031216999 & 209430157488675 \\ 139620104992450 & 305229188705674 \end{bmatrix} \quad (21)$$

Inverse einer Matrix

```
> P^(-1);
```

$$\begin{bmatrix} -2 & \frac{3}{2} \\ 1 & -\frac{1}{2} \end{bmatrix} \quad (22)$$

```
> (v.w)^(-1);
```

Error, (in rtable/Power) singular matrix

```
> U:=v.w;
```

$$U := \begin{bmatrix} 1 & 2 & 3 & 4 \\ 2 & 4 & 6 & 8 \\ 3 & 6 & 9 & 12 \\ 4 & 8 & 12 & 16 \end{bmatrix} \quad (23)$$

```
> U[1,1]:=0; U[2,2]:=0; U[3,3]:=0; U[4,4]:=0;
```

$$U_{1,1} := 0$$
$$U_{2,2} := 0$$
$$U_{3,3} := 0$$
$$U_{4,4} := 0 \quad (24)$$

```
> U; U^(-1);
```

$$\begin{bmatrix} 0 & 2 & 3 & 4 \\ 2 & 0 & 6 & 8 \\ 3 & 6 & 0 & 12 \\ 4 & 8 & 12 & 0 \end{bmatrix}$$

(25)

$$\begin{bmatrix} -\frac{2}{3} & \frac{1}{6} & \frac{1}{9} & \frac{1}{12} \\ \frac{1}{6} & -\frac{1}{6} & \frac{1}{18} & \frac{1}{24} \\ \frac{1}{9} & \frac{1}{18} & -\frac{2}{27} & \frac{1}{36} \\ \frac{1}{12} & \frac{1}{24} & \frac{1}{36} & -\frac{1}{24} \end{bmatrix} \quad (25)$$

```
> U.U^(-1);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (26)$$

SYSTEME LINEARER GLEICHUNGEN

```
> restart;
> with(LinearAlgebra):
> A:=Matrix(4,5,[[1,2,3,4,5],[6,7,8,9,10],[11,12,13,14,15],[16,17,
  18,19,20]]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 & 4 & 5 \\ 6 & 7 & 8 & 9 & 10 \\ 11 & 12 & 13 & 14 & 15 \\ 16 & 17 & 18 & 19 & 20 \end{bmatrix} \quad (27)$$

```
> b:=<1,2,3,4>;
```

$$b := \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \end{bmatrix} \quad (28)$$

Ab ist die "erweiterte Matrix"

```
> Ab:=<A|b>;
```

$$Ab := \begin{bmatrix} 1 & 2 & 3 & 4 & 5 & 1 \\ 6 & 7 & 8 & 9 & 10 & 2 \\ 11 & 12 & 13 & 14 & 15 & 3 \\ 16 & 17 & 18 & 19 & 20 & 4 \end{bmatrix} \quad (29)$$

Mit dem Befehl ReducedRowEchelonForm (Ab) wird die Zeilenstufenform von Ab berechnet:

```
> PAb:=ReducedRowEchelonForm(Ab);
```

$$PAb := \begin{bmatrix} 1 & 0 & -1 & -2 & -3 & -\frac{3}{5} \\ 0 & 1 & 2 & 3 & 4 & \frac{4}{5} \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix} \quad \textbf{(30)}$$

Mit dem Befehl NullSpace(A) wird eine Basis des Lösungsraums des durch A gegebenen homogenen Systems linearer Gleichungen berechnet:

```
> NullSpace(A);
```

$$\left\{ \begin{bmatrix} 3 \\ -4 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 2 \\ -3 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ -2 \\ 1 \\ 0 \\ 0 \end{bmatrix} \right\} \quad \textbf{(31)}$$

`> LinearSolve(A,b, free='c');` (die Lösungsmenge ist die Menge dieser Spalten, wobei die Zahlen c_i beliebig gewählt werden können)

$$\begin{bmatrix} -\frac{3}{5} + c_3 + 2\,c_4 + 3\,c_5 \\ \frac{4}{5} - 2\,c_3 - 3\,c_4 - 4\,c_5 \\ c_3 \\ c_4 \\ c_5 \end{bmatrix} \quad \textbf{(32)}$$

`> LinearSolve(Ab,free='c');` (wenn nur eine Matrix eingegeben wird, wird die letzte Spalte als Spalte der erweiterten Matrix interpretiert)

$$\begin{bmatrix} -\frac{3}{5} + c_3 + 2\,c_4 + 3\,c_5 \\ \frac{4}{5} - 2\,c_3 - 3\,c_4 - 4\,c_5 \\ c_3 \\ c_4 \\ c_5 \end{bmatrix} \quad \textbf{(33)}$$

```
> LinearSolve(PAb, free='c');
```

$$\begin{bmatrix} -\frac{3}{5}+c_3+2\,c_4+3\,c_5 \\ \frac{4}{5}-2\,c_3-3\,c_4-4\,c_5 \\ c_3 \\ c_4 \\ c_5 \end{bmatrix} \tag{34}$$

```
> Rank(A);
```

$$2 \tag{35}$$

Elementare Umformungen

`> Ab1:=Pivot(Ab,1,1);` (Durch elementare Umformungen wird erreicht, dass in der ersten Spalte der Matrix Ab in allen Zeilen außer der ersten 0 steht).

$$Ab1 := \begin{bmatrix} 1 & 2 & 3 & 4 & 5 & 1 \\ 0 & -5 & -10 & -15 & -20 & -4 \\ 0 & -10 & -20 & -30 & -40 & -8 \\ 0 & -15 & -30 & -45 & -60 & -12 \end{bmatrix} \tag{36}$$

```
> Ab2:=Pivot(Ab1,2,2);
```

$$Ab2 := \begin{bmatrix} 1 & 0 & -1 & -2 & -3 & -\frac{3}{5} \\ 0 & -5 & -10 & -15 & -20 & -4 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix} \tag{37}$$

`> RowOperation(Ab2,2,-1/5);` (Die zweite Zeile der Matrix Ab2 wird mit -1/5 multipliziert).

$$\begin{bmatrix} 1 & 0 & -1 & -2 & -3 & -\frac{3}{5} \\ 0 & 1 & 2 & 3 & 4 & \frac{4}{5} \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix} \tag{38}$$

Ein System linearer Gleichungen mit genau einer Lösung:

```
> M := <<1,1,1,4>|<1,1,-2,1>|<3,1,1,8>|<-1,1,-1,-1>>;
```

$$M := \begin{bmatrix} 1 & 1 & 3 & -1 \\ 1 & 1 & 1 & 1 \\ 1 & -2 & 1 & -1 \\ 4 & 1 & 8 & -1 \end{bmatrix} \tag{39}$$

```
> LinearSolve(M,<0,1,1,0>);
```

$$\begin{bmatrix} \frac{25}{6} \\ \frac{4}{3} \\ -\frac{5}{2} \\ -2 \end{bmatrix} \tag{40}$$

```
> M^(-1);
```

$$\begin{bmatrix} \frac{13}{6} & \frac{5}{2} & \frac{5}{3} & -\frac{4}{3} \\ \frac{4}{3} & 1 & \frac{1}{3} & -\frac{2}{3} \\ -\frac{3}{2} & -\frac{3}{2} & -1 & 1 \\ -2 & -1 & -1 & 1 \end{bmatrix} \tag{41}$$

```
> M^(-1).<0,1,1,0>;
```

$$\begin{bmatrix} \frac{25}{6} \\ \frac{4}{3} \\ -\frac{5}{2} \\ -2 \end{bmatrix} \tag{42}$$

```
> ReducedRowEchelonForm(M);
```

(43)

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(43)}$$

>